' {$STAMP BS2}

' {$PBASIC 2.5}

'~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

'========== assign pins

  ReceivePin  PIN 0

  SendPin     PIN 1

  LeftMotor   PIN 12

  RightMotor  PIN 13

  RearBumper  PIN 7

  RightBumper PIN 8

  FrontBumper PIN 9

  LeftBumper  PIN 10

'==================Declare necessary variable

  SENSORS     VAR Nib

  Direction   VAR Byte

  T           VAR Byte

  bData VAR Byte(2)    'Received data buffer from RobotBASIC

                       'byte 0 is the command code and byte 1

                       'is a parameter for the command.

  bBuffer VAR Byte(5)  'Buffer of data to be sent to RobotBASIC

                       'byte 0 is the bumper sensor. Byte 1 is the

                       'infrared sensors. Byte 2 is the line sensors.

                       'Byte 3 and 4 is a 16 bit integer for the data

                       'to be returnd as the result of some commands.

                       'Byte 3 is the MSB, byte 4 is the LSB.

                       'See RobotBASIC help for more details.

'==============================================================

Main:

  PAUSE 1000   'WAIT FOR the eb500 radio TO be ready

  DO

    SERIN ReceivePin,84,[STR bData\2] 'receive command and its

                                      'parameter from RobotBASIC

    IF bData(0) = 3  THEN

       GOSUB rLocate

    ELSEIF bData(0) = 6 THEN

       Direction = 1

       GOSUB rForward

    ELSEIF bData(0) = 7 THEN

       Direction = 0

       GOSUB rForward

    ELSEIF bData(0) = 12 THEN

       Direction = 1

       GOSUB rTurn

    ELSEIF bData(0) = 13 THEN

       Direction = 0

       GOSUB rTurn

    ELSEIF bData(0) = 24 THEN

       GOSUB rCompass

    ELSEIF bData(0) = 48 THEN

       Direction = 1

       GOSUB rLook

    ELSEIF bData(0) = 49 THEN

       Direction = 0

       GOSUB rLook

    ELSEIF bData(0) = 96 THEN

       GOSUB rBeacon

    ELSEIF bData(0) = 192 THEN

       Direction = 1

       GOSUB rRange

    ELSEIF bData(0) = 193 THEN

       Direction = 0

       GOSUB rRange

    ELSEIF bData(0) = 129 THEN

       GOSUB rPen

    ELSEIF bData(0) = 36 THEN

       GOSUB rSpeed

    ELSEIF bData(0) = 66 THEN

       GOSUB rGPS

    ELSE

       GOSUB NotRecognized

    ENDIF

    SEROUT SendPin,84,[STR bBuffer\5]  'Send back 5 bytes to RobotBASIC

  LOOP

END

'=============================================================

rLocate:

     DEBUG "rLocate ",DEC bData(1),CR

     GOSUB ReadSensors

RETURN

'=============================================================

rForward:

    ' DEBUG "rForward ",DEC bData(1),CR

    ' move the number of units specified

    ' by the variable bData(1)

    FOR T = 1 TO bData(1)

        ' This code moves our robot about

        ' 1/4 inch

        IF Direction = 1 THEN

            PULSOUT LeftMotor,775

            PULSOUT RightMotor,725

        ELSE

            PULSOUT LeftMotor,725

            PULSOUT RightMotor,775

        ENDIF

        PAUSE 10

    NEXT

    GOSUB ReadSensors

RETURN

'=============================================================

rTurn:

    ' DEBUG "rTurn ",DEC bData(1),CR

    ' This subroutine will turn

    ' the number of units specified

    ' by the variable bData(1)

    IF Direction = 1 THEN Direction = 30

    FOR T = 1 TO  bData(1)

        ' The following code turns our

        ' robot about 1 degrees

        PULSOUT LeftMotor,735+Direction

        PULSOUT RightMotor,735+Direction

        PAUSE 10

    NEXT

    GOSUB ReadSensors

RETURN

'==============================================================

rCompass:

     DEBUG "rCompass ",DEC bData(1),CR

     GOSUB ReadSensors

     bBuffer(3) = 0

     bBuffer(4) = 220    'value for testing

     RANDOM bBuffer(4)

RETURN

'=============================================================

rLook:

     DEBUG "rLook ",DEC bData(1),CR

     GOSUB ReadSensors

     bBuffer(3) = 0

     bBuffer(4) = 210  'value for testing

RETURN

'=============================================================

rBeacon:

     DEBUG "rBeacon ",DEC bData(1),CR

     GOSUB ReadSensors

     bBuffer(3) = 0

     bBuffer(4) = 50   'value for testing

RETURN

'=============================================================

rRange:

     DEBUG "rRange ",DEC bData(1),CR

     GOSUB ReadSensors

     bBuffer(3) = 0

     bBuffer(4) = 100     'value for testing

RETURN

'=============================================================

rPen:

     DEBUG "rPen ",DEC bData(1),CR

     GOSUB ReadSensors

RETURN

'=============================================================

rSpeed:

     DEBUG "rSpeed ",DEC bData(1),CR

     GOSUB ReadSensors

RETURN

'=============================================================

rGPS:

     DEBUG "rGPS ",DEC bData(1),CR

     bBuffer(0) = 1

     bBuffer(1) = 210     'values for testing

     bBuffer(2) = 2

     bBuffer(3) = 220     'values for testing

     bBuffer(4) = 0

RETURN

'=============================================================

NotRecognized:

    DEBUG "Not A Command ",DEC bData(0),"  ",DEC bData(1),CR

    bBuffer(0) = 255   'set all returned values to -1

    bBuffer(1) = 255

    bBuffer(2) = 255

    bBuffer(3) = 255

    bBuffer(4) = 255

RETURN

'=============================================================

ReadSensors:

   'read bumper sensors

   SENSORS.BIT0 = IN7

   SENSORS.BIT1 = IN8

   SENSORS.BIT2 = IN9

   SENSORS.BIT3 = IN10

   bBuffer(0) = SENSORS

   'read infrared sensors

   'none for now

   bBuffer(1) = 0

   'read Line sensors

   'none of now

   bBuffer(2) = 0

   'set the last two bytes to zero

   bBuffer(3) = 0

   bBuffer(4) = 0

RETURN

'=============================================================

Figure 4: The BS2 shell program for carrying out the communications protocol with RobotBASIC.
